Wave equation

```
ClearAll["Global`*"]

Off[General::spell, General::spell1]
```

Finding the solution of the wave equation $\partial_{x,x}\Psi[x,y,z,t]-\frac{1}{c^2}\partial_{t,t}\Psi[x,y,z,t]=0$

```
Clear["Global`*"];
```

The form of the 1D wave equation is

```
equnde1D = ∂x,x Ψ[x, t] == ∂t,t Ψ[x, t] / c^2
```

$$\Psi^{(2,0)}[x,t] == \frac{\Psi^{(0,2)}[x,t]}{c^2}$$

The solution of the above equation is

```
DSolve[equnde1D, Ψ[x, t], {x, t}]
```

$$\left\{\left\{\Psi[x,t]\to C[1]\left[t-\frac{\sqrt{c^2}\,x}{c^2}\right]+C[2]\left[t+\frac{\sqrt{c^2}\,x}{c^2}\right]\right\}\right\}$$

Rewriting the equation for the 3D case

```
equnde3D = ∂x,x Ψ[x, y, z] + ∂y,y Ψ[x, y, z] + ∂z,z Ψ[x, y, z] == -ω^2 Ψ[x, y, z] / c^2
```

$$\Psi^{(0,0,2)}[x,y,z]+\Psi^{(0,2,0)}[x,y,z]+\Psi^{(2,0,0)}[x,y,z] == -\frac{\omega^2\,\Psi[x,y,z]}{c^2}$$

DSolve is not able to solve the problem because you can lose some solutions

```
DSolve[equnde3D, Ψ[x, y, z], {x, y, z}]
```

$$\text{DSolve}\left[\Psi^{(0,0,2)}[x, y, z] + \Psi^{(0,2,0)}[x, y, z] + \Psi^{(2,0,0)}[x, y, z] == -\frac{\omega^2 \Psi[x, y, z]}{c^2}, \Psi[x, y, z], \{x, y, z\}\right]$$

Asking for the package where DSolve is

```
Needs["Calculus`DSolveIntegrals`"]
```

```
DSolve[equnde3D, Ψ[x, y, z], {x, y, z}]
```

$$\text{DSolve}\left[\Psi^{(0,0,2)}[x, y, z] + \Psi^{(0,2,0)}[x, y, z] + \Psi^{(2,0,0)}[x, y, z] == -\frac{\omega^2 \Psi[x, y, z]}{c^2}, \Psi[x, y, z], \{x, y, z\}\right]$$

Using the variable separation method, we are shearching for of solution like: Ψ[x,y,z]=X[x]*Y[y]*Z[z],for a given frequency

```
sep3 = equnde3D /. Ψ[x, y, z] → X[x] * Y[y] * Z[z] /.
  D[Ψ[x, y, z], {x, nx_}, {y, ny_}, {z, nz_}] →
   D[X[x] * Y[y] * Z[z], {x, nx}, {y, ny}, {z, nz}]
```

$$Y[y]\, Z[z]\, X''[x] + X[x]\, Z[z]\, Y''[y] + X[x]\, Y[y]\, Z''[z] == -\frac{\omega^2 X[x]\, Y[y]\, Z[z]}{c^2}$$

Dividing equation from Step 3 to X[x]*Y[y]*Z[z]

```
sep31 = Thread[1 / (X[x] * Y[y] * Z[z]) # &[sep3], Equal] // Simplify
```

$$\frac{\omega^2}{c^2} + \frac{X''[x]}{X[x]} + \frac{Y''[y]}{Y[y]} + \frac{Z''[z]}{Z[z]} == 0$$

```
sep31[[1, 4]]
```

$$\frac{Z''[z]}{Z[z]}$$

But

```
eqz = sep31[[1, 4]] == -kz^2
```

$$\frac{Z''[z]}{Z[z]} == -kz^2$$

Semnification of **ToRules[expr]**= takes logical combinations of equations, in the form generated by Roots and Reduce, and converts them to lists of rules, of the form produced by **Solve**.

```
eqxy = sep31 /. ToRules[eqz]
```

$$-kz^2 + \frac{\omega^2}{c^2} + \frac{X''[x]}{X[x]} + \frac{Y''[y]}{Y[y]} == 0$$

```
eqy = eqxy[[1, 4]] == -ky^2
```

$$\frac{Y''[y]}{Y[y]} == -ky^2$$

```
eqx = eqxy /. ToRules[eqy]
```

$$-ky^2 - kz^2 + \frac{\omega^2}{c^2} + \frac{X''[x]}{X[x]} == 0$$

Solving the equation in X[x], Y[y] si Z[z]

```
DSolve[eqx, X[x], x]
```

$$\left\{\left\{X[x] \to C[1]\,\text{Cos}\left[\frac{x\sqrt{-c^2\,ky^2 - c^2\,kz^2 + \omega^2}}{c}\right] + C[2]\,\text{Sin}\left[\frac{x\sqrt{-c^2\,ky^2 - c^2\,kz^2 + \omega^2}}{c}\right]\right\}\right\}$$

```
DSolve[eqy, Y[y], y]
```

$$\{\{Y[y] \to C[1]\,\text{Cos}[ky\,y] + C[2]\,\text{Sin}[ky\,y]\}\}$$

```
DSolve[eqz, Z[z], z]
```

$$\{\{Z[z] \to C[1]\,\text{Cos}[kz\,z] + C[2]\,\text{Sin}[kz\,z]\}\}$$

```
eqx[[1, 1]]
```

$$-ky^2$$

```
eqx[[1, 2]]
```

$$-kz^2$$

```
eqx[[1, 3]]
```

$$\frac{\omega^2}{c^2}$$

```
xrule = eqx[[1, 1]] + eqx[[1, 2]] + eqx[[1, 3]] → kx^2
```

$$-ky^2 - kz^2 + \frac{\omega^2}{c^2} \to kx^2$$

Solutions are:

```
sx = DSolve[eqx /. xrule, X[x], x] /. C → Cx
sy = DSolve[eqy, Y[y], y] /. C → Cy
sz = DSolve[eqz, Z[z], z] /. C → Cz
```

```
{{X[x] → Cos[kx x] Cx[1] + Sin[kx x] Cx[2]}}
```

```
{{Y[y] → Cos[ky y] Cy[1] + Sin[ky y] Cy[2]}}
```

```
{{Z[z] → Cos[kz z] Cz[1] + Sin[kz z] Cz[2]}}
```

General solution is

```
Ψ[x_, y_, z_] := X[x] * Y[y] * Z[z] /. sx[[1]] /. sy[[1]] /. sz[[1]]
```

```
Ψ[x, y, z]
```

```
(Cos[kx x] Cx[1] + Sin[kx x] Cx[2])
 (Cos[ky y] Cy[1] + Sin[ky y] Cy[2]) (Cos[kz z] Cz[1] + Sin[kz z] Cz[2])
```

Initial and limits condition will able us to determine the coefficients :